# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Allan McRae <mcrae_allan@hotmail.com>
# Contributor: François Charette <firmicus ατ gmx δοτ net>

pkgname=perl-unicode-utf8simple
pkgver=1.06
pkgrel=8
pkgdesc="Conversions to/from UTF8 from/to characterse"
arch=('any')
url="https://search.cpan.org/dist/Unicode-UTF8simple"
license=('GPL' 'PerlArtistic')
depends=('perl>=5.10.0')
options=('!emptydirs')
source=(https://www.cpan.org/authors/id/G/GU/GUS/Unicode-UTF8simple-$pkgver.tar.gz)
sha256sums=('e1249ac08daceb8a83c170c00810001f15d73418e9595711bee663d0130c0fe5')

build() {
  cd  "$srcdir"/Unicode-UTF8simple-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd  "$srcdir"/Unicode-UTF8simple-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
